


# Programmierung von ImageJ-Macros


# Was ist ImageJ?

- Bildbearbeitungs- und Bildverarbeitungsprogramm
- Plattformunabhängig (Java)
- Open Source
- Erweiterbar durch Plugins
  - Java Plugins
  - Macro Language






# ImageJ



# Image Funktionen



# Process Funktionen



# „Hello world“ Example

```
print("Hello world");
```

- Externer Texteditor (nach belieben)
- Speichern als Textdatei
- Start über „Macros → Run Macro“

# Beispiel 2

```
filename = "...";  
newname = "...";  
  
open(filename);  
run("8-bit Color", "number=256");  
save("PNG", newname);
```

- Öffnen
- Konvertieren
- Speichern

# Beispiel 2

```
PATH = "...";  
  
list = getFileList(PATH);  
for (i=0; i<list.length; i++) {  
    open(PATH+list[i]);  
    run("8-bit Color", "number=256");  
    save("PNG", PATH+list[i]);  
    close();  
}
```

- Öffnen
- Konvertieren
- Speichern

- + Schleife über alle Bilder in einem Verzeichnis



# Beispiel 2

- Öffnen
- Konvertieren
- Speichern

```
dir1 = getDirectory("Choose Source  
Directory");  
dir2 = getDirectory("Choose Destination  
Directory");  
  
list = getFileList(dir1);  
for (i=0; i<list.length; i++) {  
    open(dir1+list[i]);  
    run("8-bit Color", "number=256");  
    save("PNG", dir2+list[i]);  
    close();  
}
```

- + Auswahlmöglichkeit des Quell- und Zielverzeichnis

# Beispiel 2

```
dir1 = getDirectory("Choose Source Directory");
format = getFormat();
dir2 = getDirectory("Choose Destination Directory");
list = getFileList(dir1);
setBatchMode(true);
for (i=0; i<list.length; i++) {
    showProgress(i+1, list.length);
    open(dir1+list[i]);
    if (format=="8-bit TIFF" || format=="GIF")
        convertTo8Bit();
    saveAs(format, dir2+list[i]);
    close();
}

function getFormat() {
    formats = newArray("TIFF", "8-bit TIFF",
        "JPEG", "GIF", "PNG", "PGM", "BMP");
    Dialog.create("Batch Convert");
    Dialog.addChoice("Convert to: ", formats, "TIFF");
    Dialog.show();
    return Dialog.getChoice();
}

function convertTo8Bit() {
    if (bitDepth==24)
        run("8-bit Color", "number=256");
    else
        run("8-bit");
}
```

- + Auswahlmöglichkeit Zielformats

# Referenzen

- Documentation:

<http://rsbweb.nih.gov/ij/docs/index.html>

- ImageJ User Guide:

<http://rsbweb.nih.gov/ij/docs/user-guide.pdf>

- Macro Language:

[http://rsbweb.nih.gov/ij/docs/macro\\_reference\\_guide.pdf](http://rsbweb.nih.gov/ij/docs/macro_reference_guide.pdf)

- ImageJ Information and Documentation Portal (WIKI)

<http://imagejdocu.tudor.lu/>

- Developer Resources:

<http://rsbweb.nih.gov/ij/developer/index.html>